NAG Fortran Library Routine Document

F08UPF (ZHBGVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08UPF (ZHBGVX) computes all the eigenvalues and, optionally, the eigenvectors of a complex generalized Hermitian-definite banded eigenproblem, of the form

$$Az = \lambda Bz,$$

where A and B are Hermitian and banded, and B is also positive-definite. Eigenvalues and eigenvectors can be selected by specifying either all eigenvalues, a range of values or a range of indices for the desired eigenvalues.

2 Specification

```
 SUBROUTINE F08UPF (JOBZ, RANGE, UPLO, N, KA, KB, AB, LDAB, BB, LDBB, Q,
1                   LDQ, VL, VU, IL, IU, ABSTOL, M, W, Z, LDZ, WORK,
2                   RWORK, IWORK, JFAIL, INFO)
 INTEGER            N, KA, KB, LDAB, LDBB, LDQ, IL, IU, M, LDZ, IWORK(*),
1                   JFAIL(*), INFO
 double precision   VL, VU, ABSTOL, W(*), RWORK(*)
 complex*16         AB(LDAB,*), BB(LDBB,*), Q(LDQ,*), Z(LDZ,*), WORK(*)
 CHARACTER*1        JOBZ, RANGE, UPLO
```

The routine may be called by its LAPACK name ***zhbgvx***.

3 Description

The generalized Hermitian-definite band problem

$$Az = \lambda Bz$$

is first reduced to a standard band Hermitian problem

$$Cx = \lambda x,$$

where C is a Hermitian band matrix, using Wilkinson's modification to Crawford's algorithm (see Crawford (1973) and Wilkinson (1977)). The Hermitian eigenvalue problem is then solved for the required eigenvalues and eigenvectors, and the eigenvectors are then backtransformed to the eigenvectors of the original problem.

The eigenvectors are normalized so that

$$z^{\mathrm{H}}Az = \lambda \quad \text{and} \quad z^{\mathrm{H}}Bz = 1.$$

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Crawford C R (1973) Reduction of a band-symmetric generalized eigenvalue problem *Comm. ACM* **16** 41–44

Demmel J W and Kahan W (1990) Accurate singular values of bidiagonal matrices *SIAM J. Sci. Statist. Comput.* **11** 873–912

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1977) Some recent advances in numerical linear algebra *The State of the Art in Numerical Analysis* (ed D A H Jacobs) Academic Press

5 Parameters

1: JOBZ – CHARACTER*1 *Input*

On entry: if JOBZ = 'N', compute eigenvalues only.

If JOBZ = 'V', compute eigenvalues and eigenvectors.

Constraint: JOBZ = 'N' or 'V'.

2: RANGE – CHARACTER*1 *Input*

On entry: if RANGE = 'A', all eigenvalues will be found.

If RANGE = 'V', all eigenvalues in the half-open interval $(\mathrm{VL}, \mathrm{VU}]$ will be found.

If RANGE = 'I', the ILth to IUth eigenvalues will be found.

3: UPLO – CHARACTER*1 *Input*

On entry: if UPLO = 'U', the upper triangles of A and B are stored.

If UPLO = 'L', the lower triangles of A and B are stored.

4: N – INTEGER *Input*

On entry: n, the order of the matrices A and B.

Constraint: $\mathrm{N} \geq 0$.

5: KA – INTEGER *Input*

On entry: ka, the number of superdiagonals of the matrix A if UPLO = 'U', or the number of subdiagonals if UPLO = 'L'.

Constraint: $\mathrm{KA} \geq 0$.

6: KB – INTEGER *Input*

On entry: kb, the number of superdiagonals of the matrix B if UPLO = 'U', or the number of subdiagonals if UPLO = 'L'.

Constraint: $\mathrm{KB} \geq 0$.

7: AB(LDAB,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1, \mathrm{N})$.

On entry: the upper or lower triangle of the symmetric band matrix A, stored in the first $ka + 1$ rows of the array. The jth column of A is stored in the jth column of the array AB as follows:

if UPLO = 'U', $\mathrm{AB}(ka + 1 + i - j, j) = a_{ij}$ for $\max(1, j - ka) \leq i \leq j$;
if UPLO = 'L', $\mathrm{AB}(1 + i - j, j) = a_{ij}$ for $j \leq i \leq \min(n, j + ka)$.

On exit: the contents of AB are destroyed.

8: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F08UPF (ZHBGVX) is called.

Constraint: $\mathrm{LDAB} \geq \mathrm{KA} + 1$.

9: BB(LDBB,∗) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array BB must be at least $\max(1, \mathrm{N})$.

On entry: the upper or lower triangle of the Hermitian band matrix B, stored in the first $kb + 1$ rows of the array. The jth column of B is stored in the jth column of the array BB as follows:

if UPLO = 'U', $\mathrm{BB}(kb + 1 + i - j, j) = b_{ij}$ for $\max(1, j - kb) \leq i \leq j$;
if UPLO = 'L', $\mathrm{BB}(1 + i - j, j) = b_{ij}$ for $j \leq i \leq \min(n, j + kb)$.

On exit: the factor S from the split Cholesky factorization $B = S^{\mathrm{H}}S$, as returned by F08UTF (ZPBSTF).

10: LDBB – INTEGER *Input*

On entry: the first dimension of the array BB as declared in the (sub)program from which F08UPF (ZHBGVX) is called.

Constraint: LDBB $\geq$ KB + 1.

11: Q(LDQ,∗) – ***complex*16*** array *Output*

Note: the second dimension of the array Q must be at least $\max(1, \mathrm{N})$.

On exit: if JOBZ = 'V', the n by n matrix used in the reduction of $Az = \lambda Bz$ to standard form, i.e., $Cx = \lambda x$, and subsequently C to tridiagonal form.

If JOBZ = 'N', Q is not referenced.

12: LDQ – INTEGER *Input*

On entry: the first dimension of the array Q as declared in the (sub)program from which F08UPF (ZHBGVX) is called.

Constraints:

if JOBZ = 'N', LDQ ≥ 1;
if JOBZ = 'V', LDQ $\geq \max(1, \mathrm{N})$.

13: VL – ***double precision*** *Input*
14: VU – ***double precision*** *Input*

On entry: if RANGE = 'V', the lower and upper bounds of the interval to be searched for eigenvalues.

If RANGE = 'A' or 'I', VL and VU are not referenced.

Constraint: VL $<$ VU.

15: IL – INTEGER *Input*
16: IU – INTEGER *Input*

On entry: if RANGE = 'I', the indices (in ascending order) of the smallest and largest eigenvalues to be returned.

If RANGE = 'A' or 'V', IL and IU are not referenced.

Constraints:

if N = 0, IL = 1 and IU = 0;
if N > 0, $1 \leq$ IL $\leq$ IU $\leq$ N.

17: ABSTOL – ***double precision*** *Input*

On entry: the absolute error tolerance for the eigenvalues. An approximate eigenvalue is accepted as converged when it is determined to lie in an interval $[a, b]$ of width less than or equal to

$$\mathrm{ABSTOL} + \epsilon \max(|a|, |b|),$$

where ϵ is the ***machine precision***. If ABSTOL is less than or equal to zero, then $\epsilon\|T\|_1$ will be used in its place, where T is the tridiagonal matrix obtained by reducing C to tridiagonal form. Eigenvalues will be computed most accurately when ABSTOL is set to twice the underflow threshold $2 \times$ X02AMF( ), not zero. If this routine returns with INFO > 0, indicating that some eigenvectors did not converge, try setting ABSTOL to $2 \times$ X02AMF( ). See Demmel and Kahan (1990).

18: M – INTEGER *Output*

On exit: the total number of eigenvalues found.

If RANGE = 'A', M = N.

If RANGE = 'I', M = IU − IL + 1.

Constraint: $0 \leq \mathrm{M} \leq \mathrm{N}$.

19: W(∗) – ***double precision*** array *Output*

Note: the dimension of the array W must be at least $\max(1, \mathrm{N})$.

On exit: if INFO = 0, the eigenvalues in ascending order.

20: Z(LDZ,∗) – ***complex*16*** array *Output*

Note: the second dimension of the array Z must be at least $\max(1, \mathrm{N})$.

On exit: if JOBZ = 'V', then if INFO = 0, Z contains the matrix Z of eigenvectors, with the ith column of Z holding the eigenvector associated with W(i). The eigenvectors are normalized so that $Z^{\mathrm{H}}BZ = I$.

If JOBZ = 'N', Z is not referenced.

21: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08UPF (ZHBGVX) is called.

Constraints:

if JOBZ = 'V', LDZ $\geq \max(1, \mathrm{N})$;
LDZ ≥ 1 otherwise.

22: WORK(∗) – ***complex*16*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{N})$.

23: RWORK(∗) – ***double precision*** array *Workspace*

Note: the dimension of the array RWORK must be at least $\max(1, 7 \times \mathrm{N})$.

24: IWORK(∗) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, 5 \times \mathrm{N})$.

25: JFAIL(∗) – INTEGER array *Output*

Note: the dimension of the array JFAIL must be at least $\max(1, \mathrm{N})$.

On exit: if JOBZ = 'V', then if INFO = 0, the first M elements of JFAIL are zero.

If INFO > 0, JFAIL contains the indices of the eigenvectors that failed to converge.

If JOBZ = 'N', JFAIL is not referenced.

26: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$ and $i \leq$ N, then i eigenvectors failed to converge. Their indices are stored in array JFAIL. Please see ABSTOL.

If INFO $= i$ and $i >$ N, if INFO $=$ N $+ i$, for $1 \leq i \leq$ N, then F08UTF (ZPBSTF) returned INFO $= i$: B is not positive-definite. The factorization of B could not be completed and no eigenvalues or eigenvectors were computed.

7 Accuracy

If B is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and vectors may be large, although when the diagonal elements of B differ widely in magnitude the eigenvalues and eigenvectors may be less sensitive than the condition of B would suggest. See Section 4.10 of Anderson *et al.* (1999) for details of the error bounds.

8 Further Comments

The total number of floating-point operations is proportional to n^3 if JOBZ $=$ 'V' and RANGE $=$ 'A', and assuming that $n \gg k_a$, is approximately proportional to $n^2 k_a$ if JOBZ $=$ 'N'. Otherwise the number of floating-point operations depends upon the number of eigenvectors computed.

The real analogue of this routine is F08UBF (DSBGVX).

9 Example

This example finds the eigenvalues in the half-open interval $(0.0, 2.0]$, and corresponding eigenvectors, of the generalized band Hermitian eigenproblem $Az = \lambda Bz$, where

$$A = \begin{pmatrix} -1.13 & 1.94 - 2.10i & -1.40 + 0.25i & 0 \\ 1.94 + 2.10i & -1.91 & -0.82 - 0.89i & -0.67 + 0.34i \\ -1.40 - 0.25i & -0.82 + 0.89i & -1.87 & -1.10 - 0.16i \\ 0 & -0.67 - 0.34i & -1.10 + 0.16i & 0.50 \end{pmatrix}$$

and

$$B = \begin{pmatrix} 9.89 & 1.08 - 1.73i & 0 & 0 \\ 1.08 + 1.73i & 1.69 & -0.04 + 0.29i & 0 \\ 0 & -0.04 - 0.29i & 2.65 & -0.33 + 2.24i \\ 0 & 0 & -0.33 - 2.24i & 2.17 \end{pmatrix}.$$

9.1 Program Text

```
*     F08UPF Example Program Text
*     Mark 21.  NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          KAMAX, KBMAX, NMAX, MMAX
      PARAMETER        (KAMAX=5,KBMAX=5,NMAX=20,MMAX=10)
      INTEGER          LDAB, LDBB, LDQ, LDZ
      PARAMETER        (LDAB=KAMAX+1,LDBB=KBMAX+1,LDQ=NMAX,LDZ=NMAX)
      CHARACTER        UPLO
      PARAMETER        (UPLO='U')
      DOUBLE PRECISION ZERO
```

```
      PARAMETER        (ZERO=0.0D+0)
*     .. Local Scalars ..
      DOUBLE PRECISION ABSTOL, VL, VU
      INTEGER          I, IFAIL, IL, INFO, IU, J, KA, KB, M, N
*     .. Local Arrays ..
      COMPLEX *16      AB(LDAB,NMAX), BB(LDBB,NMAX), Q(LDQ,NMAX),
     +                 WORK(NMAX), Z(LDZ,MMAX)
      DOUBLE PRECISION RWORK(7*NMAX), W(NMAX)
      INTEGER          INDEX(NMAX), IWORK(5*NMAX)
*     .. External Subroutines ..
      EXTERNAL         X04DAF, ZHBGVX
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08UPF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KA, KB
      IF (N.LE.NMAX .AND. KA.LE.KAMAX .AND. KB.LE.KBMAX) THEN
*
*        Read the lower and upper bounds of the interval to be searched,
*        and read the upper or lower triangular parts of the matrices A
*        and B from data file
*
         READ (NIN,*) VL, VU
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AB(KA+1+I-J,J),J=I,MIN(N,I+KA)),I=1,N)
            READ (NIN,*) ((BB(KB+1+I-J,J),J=I,MIN(N,I+KB)),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AB(1+I-J,J),J=MAX(1,I-KA),I),I=1,N)
            READ (NIN,*) ((BB(1+I-J,J),J=MAX(1,I-KB),I),I=1,N)
         END IF
*
*        Set the absolute error tolerance for eigenvalues.  With ABSTOL
*        set to zero, the default value is used instead
*
         ABSTOL = ZERO
*
*        Solve the generalized symmetric eigenvalue problem
*        A*x = lambda*B*x
*
         CALL ZHBGVX('Vectors','Values in range',UPLO,N,KA,KB,AB,LDAB,
     +               BB,LDBB,Q,LDQ,VL,VU,IL,IU,ABSTOL,M,W,Z,LDZ,WORK,
     +               RWORK,IWORK,INDEX,INFO)
*
         IF (INFO.GE.0 .AND. INFO.LE.N .AND. M.LE.MMAX) THEN
*
*           Print solution
*
            WRITE (NOUT,99999) 'Number of eigenvalues found =', M
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Eigenvalues'
            WRITE (NOUT,99998) (W(J),J=1,M)
*
            IFAIL = 0
            CALL X04DAF('General',' ',N,M,Z,LDZ,'Selected eigenvectors',
     +                  IFAIL)
            IF (INFO.GT.0) THEN
               WRITE (NOUT,99999)
     +           'INFO eigenvectors failed to converge, INFO =', INFO
               WRITE (NOUT,*)
     +           'Indices of eigenvectors that did not converge'
               WRITE (NOUT,99997) (INDEX(J),J=1,M)
            END IF
         ELSE IF (INFO.GT.N .AND. INFO.LE.2*N) THEN
            I = INFO - N
            WRITE (NOUT,99996) 'The leading minor of order ', I,
     +        ' of B is not positive definite'
         ELSE IF (M.GT.MMAX) THEN
            WRITE (NOUT,99995) 'M greater than MMAX, M =', M,
```

```
     +        ', MMAX =', MMAX
         ELSE
            WRITE (NOUT,99999) 'Failure in ZHBGVX. INFO =', INFO
         END IF
       ELSE
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'NMAX too small'
       END IF
       STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (3X,(8F8.4))
99997 FORMAT (3X,(8I8))
99996 FORMAT (1X,A,I4,A)
99995 FORMAT (1X,A,I5,A,I5)
      END
```

9.2 Program Data

```
F08UPF Example Program Data

   4             2             1                          :Values of N, KA and KB

   0.0           2.0                                      :Values of VL and VU

 (-1.13, 0.00) ( 1.94,-2.10) (-1.40, 0.25)
               (-1.91, 0.00) (-0.82,-0.89) (-0.67, 0.34)
                             (-1.87, 0.00) (-1.10,-0.16)
                                           ( 0.50, 0.00) :End of matrix A

 ( 9.89, 0.00) ( 1.08,-1.73)
               ( 1.69, 0.00) (-0.04, 0.29)
                             ( 2.65, 0.00) (-0.33, 2.24)
                                           ( 2.17, 0.00) :End of matrix B
```

9.3 Program Results

```
 F08UPF Example Program Results

 Number of eigenvalues found =    2

 Eigenvalues
     0.1603  1.7712
 Selected eigenvectors
             1          2
 1       0.1908     0.0494
         0.0137    -0.0045

 2       0.1413     0.2505
         0.1012     0.4427

 3      -0.0437    -0.9705
        -0.0905     0.0679

 4      -0.2135     0.0606
         0.2880    -1.3227
```